(DEFUN Hélice ()

  (SETQ Ind (LOAD_DIALOG "c:/misdoc~1/autocad/dcl/helice-pol.dcl"))

  (NEW_DIALOG "helice" Ind)

  (IF radin0 () (SETQ radin0 10))

  (SET_TILE "radin" (RTOS radin0 2 2))

  (IF radfin0 () (SETQ radfin0 10))

  (SET_TILE "radif" (RTOS radfin0 2 2))

  (IF nv0 () (SETQ nv0 1))

  (SET_TILE "nv" (ITOA nv0))

  (IF paso0 () (SETQ paso0 1))

  (SET_TILE "pas" (RTOS paso0 2 2))

  (IF alt0 () (SETQ alt0 10))

  (SET_TILE "alt" (RTOS alt0 2 2))    

  (IF op1 () (SETQ op1 "1"))

  (IF (= op1 "1")

    (PROGN (SET_TILE "igu" "1")(Iguales))

    (PROGN (SET_TILE "dif" "1")(Diferentes))

  )

  (IF op2 () (SETQ op2 "0"))

  (SET_TILE "pre" op2) 

  (IF op3 () (SETQ op3 "1"))

  (IF (= op3 "1")

    (PROGN (SET_TILE "bpas" "1")(BPaso))

    (PROGN (SET_TILE "balt" "1")(BAltura))

  )

  (ACTION_TILE "igu" "(Iguales)")

  (ACTION_TILE "dif" "(Diferentes)")

  (ACTION_TILE "bpas" "(BPaso)")

  (ACTION_TILE "balt" "(BAltura)")

  (ACTION_TILE "accept" "(Chequear)(IF errores () (Aceptar))")

  (IF (= (START_DIALOG) 1)(Dibujar_Hélice))

)

(DEFUN Iguales ()

  (START_IMAGE "img")

  (FILL_IMAGE 0 0 (DIMX_TILE "img") (DIMY_TILE "img") 0)

  (SLIDE_IMAGE 0 0 (DIMX_TILE "img") (DIMY_TILE "img") 

  "c:/misdoc~1/autocad/dcl/helice1.sld")

  (END_IMAGE)

  (MODE_TILE "radif" 1)

)

(DEFUN Diferentes ()

  (START_IMAGE "img")

  (FILL_IMAGE 0 0 (DIMX_TILE "img") (DIMY_TILE "img") 0)

  (SLIDE_IMAGE 0 0 (DIMX_TILE "img") (DIMY_TILE "img") 

  "c:/misdoc~1/autocad/dcl/helice2.sld")

  (END_IMAGE)

  (MODE_TILE "radif" 0)

)

(DEFUN BPaso ()

  (MODE_TILE "pas" 0)

  (MODE_TILE "alt" 1)

)

(DEFUN BAltura ()

  (MODE_TILE "pas" 1)

  (MODE_TILE "alt" 0)

)   

(DEFUN Chequear ()

  (SETQ Errores nil)

  (SETQ radin0 (ATOF (GET_TILE "radin")))

  (IF (< radin0 0)

    (PROGN

      (SETQ errores T)

      (SET_TILE "error" "Radio inicial no puede ser negativo.")

      (MODE_TILE "radin" 2)

    )

  )

  (SETQ radfin0 (ATOF (GET_TILE "radif")))

  (IF (< radfin0 0)

    (PROGN

      (SETQ errores T)

      (SET_TILE "error" "Radio final no puede ser negativo.")

      (MODE_TILE "radif" 2)

    )

  )

  (SETQ nv0 (ATOI (GET_TILE "nv")))

  (IF (<= nv0 0)

    (PROGN

      (SETQ errores T)

      (SET_TILE "error" "Número de vueltas ha de ser mayor de 0.")

      (MODE_TILE "nv" 2)

    )

  )

)

(DEFUN Aceptar ()

  (SETQ paso0 (ATOF (GET_TILE "pas")))

  (SETQ alt0 (ATOF (GET_TILE "alt")))

  (SETQ op1 (GET_TILE "igu"))

  (SETQ op2 (GET_TILE "pre"))

  (SETQ op3 (GET_TILE "bpas"))

  (COND

    ((= op2 "0")(SETQ pv 8))

    ((= op2 "1")(SETQ pv 16))

    ((= op2 "2")(SETQ pv 24)) 

    ((= op2 "3")(SETQ pv 32))

  )

  (DONE_DIALOG 1)    

  (SETQ radin radin0 radfin radfin0 nv nv0 paso paso0 alt alt0)

)

(DEFUN Dibujar_Hélice (/ cen dz drad dang n z z0)

   (INITGET 1)

   (SETQ cen (GETPOINT "Centro de la hélice: "))(TERPRI)

   (IF (= op1 "1") (SETQ radfin radin))

   (IF (= op3 "0") (SETQ paso (/ alt nv)))

   (SETQ dang (/ (* 2 PI) pv))

   (SETQ dz (/ paso pv))

   (SETQ drad (/ (- radfin radin) (* pv nv )))

   (SETQ p0 (POLAR cen 0 radin))

   (SETQ z0 (CADDR p0))

   (COMMAND "_3dpoly" p0)

   (SETQ n 1)

   (REPEAT (* pv nv)

     (SETQ z (+ z0 (* dz n)))

     (SETQ xy (POLAR cen (* dang n) (+ radin (* drad n))))

     (SETQ pto (LIST (CAR xy) (CADR xy) z))

     (COMMAND pto)

     (SETQ n (1+ n))

   )

   (COMMAND)

)

(DEFUN C:Hélice ()

  (SETVAR "CMDECHO" 0)

  (SETQ Error_Dibujo *error* *error* Control_Errores)

  (Hélice)

  (SETQ *error* Error_Dibujo)

  (SETVAR "CMDECHO" 1)

  (PRIN1)

)

(DEFUN Control_Errores (Mensaje)

  (PRINC (STRCAT "Error: " Mensaje))(TERPRI)

  (SETQ *error* Error_Dibujo)

  (SETVAR "CMDECHO" 1) 

  (COMMAND)

  (PRIN1)

)
